类 推荐布局:组件容器

	@布局配置([[根布局=真,宽度=-1,高度=-1,纵向布局=真]])
	变量 线性布局1 : 线性布局
	@布局配置([[父布局=线性布局1,宽度=-1,高度DP=55,对齐方式=17,左内边距DP=18,右内边距DP=18,上内边距DP=15,下内边距DP=5]])
	变量 线性布局2 : 线性布局
	@布局配置([[父布局=线性布局2,宽度=-1,高度=-1,卡片阴影=0,圆角=75,背景颜色=-1]])
	变量 卡片布局1 : 卡片布局
	@布局配置([[父布局=卡片布局1,宽度=-1,高度=-1]])
	变量 线性布局3 : 线性布局
	@布局配置([[父布局=线性布局3,宽度DP=35,高度=-1,对齐方式=8388629,@右外边距DP=6]])
	变量 线性布局4 : 线性布局
	@布局配置([[父布局=线性布局4,宽度DP=20,高度DP=20,图片资源="../../../素材/内置/png/搜索音乐图标.png"]])
	变量 图片框1 : 图片框
	@布局配置([[父布局=线性布局3,高度=-1,对齐方式=16,字体颜色=-5460040,粗体=假,字体大小=14,内容="请输入关键词"]])
	变量 文本框1 : 文本框
	@布局配置([[父布局=线性布局3,高度=-1,对齐方式=16,字体颜色=-5460040,粗体=假,字体大小=10,内容="",@左外边距DP=8,透明度=0.53]])
	变量 文本框2 : 文本框
	@布局配置([[父布局=线性布局1,宽度=-1,高度=-1,纵向布局=真]])
	变量 线性布局6 : 线性布局
	@布局配置([[父布局=线性布局6,宽度=-1,高度DP=30,左内边距DP=18,右内边距DP=18,@上外边距DP=5,@下外边距DP=5]])
	变量 线性布局5 : 线性布局
	@布局配置([[父布局=线性布局5,宽度=-1,高度=-1]])
	变量 导航栏1 : 导航栏
	@布局配置([[父布局=线性布局6,宽度=-1,高度=-1]])
	变量 分页布局1 : 分页布局

	变量 导航标题集合 : 文本集合={"推荐","歌单","音乐馆"}
	变量 推荐1:推荐页推荐布局
	变量 歌单1:推荐页歌单布局
	变量 音乐1:推荐页音乐布局

	变量 时钟1:时钟
	变量 网络工具1:网络工具
	变量 搜索预选标题集合:文本集合
	变量 搜索预选备注集合:文本集合
	变量 当前搜索预选索引:整数
	变量 透明动画1 : 透明动画扩增?

	变量 音乐播放器1:音乐播放器

	事件 推荐布局:创建完毕()
		订阅事件()
		分页布局1.添加页面(推荐1)
		分页布局1.添加页面(歌单1)
		分页布局1.添加页面(音乐1)
		事件 导航栏1:导航文本样式(索引 : 整数,导航文本 : 文本) : 导航文本样式
			变量 文本样式 : 导航文本样式_渐变缩放
			文本样式.字体大小 = 17
			文本样式.粗体 = 真
			文本样式.缩小度 = 0.9
			文本样式.颜色_选中 = -12687655
			文本样式.颜色_未选中 = 深灰色
			返回 文本样式
		结束 事件
		事件 导航栏1:导航条样式() : 导航条样式
			变量 条样式 : 导航条样式_直线渐变
			条样式.模式 = 导航条宽模式.导航文本宽度
			条样式.圆角 = 6
			条样式.滚动长度 = 2.0
			条样式.渐变颜色组 = {-12687655}
			返回 条样式
		结束 事件
		导航栏1.绑定分页布局(分页布局1)
		导航栏1.置文本集(导航标题集合)
		网络工具1.取网页源码(公用数据.搜索预选接口)
		透明动画1 = 透明动画扩增.创建动画(0,1)
		透明动画1.播放时间 = 500
		透明动画1.保持最终状态 = 假
	结束 事件	
	事件 网络工具1:取网页源码结束(结果 : 文本,内容 : 字节[],cookie : 文本)
		变量 搜索预选词对象:JSON对象=结果
		变量 搜索预选词数组:JSON数组=搜索预选词对象.取JSON数组("content")
		循环(搜索预选数组索引, 0, 搜索预选词数组.长度)
			搜索预选标题集合.添加成员(搜索预选词数组.取JSON对象(搜索预选数组索引).取文本("query_word"))
			搜索预选备注集合.添加成员(搜索预选词数组.取JSON对象(搜索预选数组索引).取文本("desc"))
		结束 循环
		当前搜索预选索引=0
		文本框1.内容=搜索预选标题集合[当前搜索预选索引]
		文本框2.内容=搜索预选备注集合[当前搜索预选索引]
		时钟1.时钟周期 = 3000
	结束 事件
	事件 网络工具1:取网页源码失败()
		变量 搜索预选词对象:JSON对象=读入资源文件(取安卓窗口(),"json/搜索预选关键词.json")
		变量 搜索预选词数组:JSON数组=搜索预选词对象.取JSON数组("content")
		循环(搜索预选数组索引, 0, 搜索预选词数组.长度)
			搜索预选标题集合.添加成员(搜索预选词数组.取JSON对象(搜索预选数组索引).取文本("query_word"))
			搜索预选备注集合.添加成员(搜索预选词数组.取JSON对象(搜索预选数组索引).取文本("desc"))
		结束 循环
		当前搜索预选索引=取随机数(0,搜索预选标题集合.长度-1)
		文本框1.内容=搜索预选标题集合[当前搜索预选索引]
		文本框2.内容=搜索预选备注集合[当前搜索预选索引]
		时钟1.时钟周期 = 3000
	结束 事件
	事件 时钟1:周期事件()
		如果 当前搜索预选索引==搜索预选标题集合.长度-1 则
			当前搜索预选索引=0
		否则
			当前搜索预选索引=当前搜索预选索引+1
		结束 如果
		文本框1.播放动画(透明动画1)
	结束 事件
	事件 文本框1:动画开始播放()
		文本框1.内容=搜索预选标题集合[当前搜索预选索引]
		文本框2.内容=搜索预选备注集合[当前搜索预选索引]
	结束 事件
结束 类